2

REPRESENTATION OF CURVES IN EXPLICIT FORM

In mathematics the graphical representation of a function is really important. The graphical representation of a curve helps when calculating areas, volumes, maximum and minimum values, etc. We will use the bidimensional system OXY of rectangular coordinates to make the graphical representation of a explicit form function. The representation of the points (x,f(x)) is the graph of the function f.



2.1. Using the command Plot

This command is used to make the graphical representation of an explicit function y=f(x) in a rectangular bidimensional system OXY. The representation of the points (x,f(x)) is the graph of the function f.

▼ Plot

⋆ Plot [function , {x, xmin, xmax}]

```
f1[x_] = x^2
```

x^2

x^2

```
f2[x_] = x^3
```

x^3

x^3

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Plot[f1[x], {x, -2, 2}]
```



⋆ Plot [{function1, function2,...., functionn} , {x, xmin, xmax}]

It is possible to make the graphical representation of many functions using the same axes

```
Plot[{f1[x], f2[x]}, {x, -2, 2}]
```



2.2. Options of the command Plot

▼ Style options

It is possible to change the default values of the command Plot. That is to say, it is possible to specify the colour of a graph, its thickness, etc. The different options must be separated by commas.

⋆ PlotStyle → colour

It plots the graphic of the function using the specified colour

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Plot[f1[x], {x, -2, 2}, PlotStyle → RGBColor[ 0, 0, 1]]
```



★ PlotStyle → {RGBColor[, ,], Thickness[*n*], Dashing[n]}

It plots the graphic of the function using the specified line thickness, line style, and so on. These values are specified using the variable *n*.

```
Plot[f1[x], {x, -2, 2}, PlotStyle → {RGBColor[0, 0.5, 1], Thickness[0.01], Dashing[0.02]}]
```



★ PlotStyle → {{RGBColor [, ,], Thickness[n]},{RGBColor[, ,],Thickness[*n*],}}

This command assigns to each of the curves the specified colour, thickness, etc.

```
Plot[{f1[x], f2[x]}, {x, -2, 2}, PlotStyle →
  {{RGBColor[ 1, 0, 0], Thickness[0.01]}, {RGBColor[ 0, 0, 1], Thickness[0.01]}}]
```



★ PlotRange → {y1,y2}, PlotRange → {{x1, x2},{y1 , y2}}

It is to define the range in which the curve is going to be plotted.

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Plot[{f1[x], f2[x]}, {x, -2, 2}, PlotRange → {-4, 4}]
```




```
Plot[{f1[x], f2[x]}, {x, -2, 2}, PlotRange → {{-2, 2}, {-2, 4}}]
```



⋆ AspectRatio → Automatic

It is to define the relation between the height and the width of the graph.

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Plot[{f1[x], f2[x]}, {x, -2, 2}, PlotStyle →
  {{RGBColor[1, 0, 0], Thickness[0.01]}, {RGBColor[0, 0, 1], Thickness[0.01]}},
 PlotRange → {{-2, 2}, {-2, 4}}, AspectRatio → Automatic]
```



▼ Labels

★ **Epilog → {Text1[Style[text, colour, size], coordinates],Text2[Style[text, ...], coordinates]}**

```
Plot[{f1[x], f2[x]}, {x, -2, 2},
 PlotStyle → {{Blue, Thickness[0.01]}, {Magenta, Thickness[0.01]}},
 PlotRange → {{-2, 2}, {-2, 3}}, Epilog → {Text[Style["y=x^2", 14, Blue, Bold], {-0.5, 1}],
   Text[Style["y=x^3", 14, Magenta, Bold], {-0.5, -1}]}]
```



★ **PlotLabel → name, Ticks->{{x1, x2, ..},{y1, y2, ...}}, AxesLabel → {Name of the OX axis, Name of the OY axis}**

It writes above the graph the name that we have specified, and it inserts the names of the axes.

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Plot[{Sin[x], Cos[x]}, {x, 0, 2 π},
 PlotStyle → {{Blue, Thickness[0.010], Dashing[0.015]}, {Purple, Thickness[0.010]}},
 PlotLabel → Style["sinx and cosx functions", 18, Bold, Magenta],
 Ticks → {{0, Pi / 2, Pi, 3 Pi / 2, 2 Pi}, {-1, 1}},
 AxesLabel → {Style["OX", 12, Bold], Style["OY", 12, Bold]}]
```



▼ Some other options of the command Plot

★ Non-continuous functions: Exclusions

$$f[x\_] = \frac{x^3}{x^2 - 4};$$

```
Plot[f[x], {x, -4, 4}, PlotStyle -> {Blue, Thickness[0.01]},
 Exclusions → {x^2 - 4 == 0}, ExclusionsStyle → Dashing[0.01]]
```



★ Axes → False

To eliminate the axes

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Plot[Sin[x], {x, -2 π, 2 π}, Axes → False]
```



★ Filling → (Axis/Bottom/Top)

To colour the region between the axis and the function

```
Plot[Sin[x], {x, -2 π, 2 π}, Filling → Axis]
```



★ Filling → {*n1* → {*n2*}}

It colours the space between two functions

```
Plot[{Sin[x], Cos[x]}, {x, 0, 2 π}, Filling → {1 → {2}}]
```



★ Background → *colour*

It colours the background of the graph

★ Frame → True

It puts a frame to the graph

E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
Plot[Sin[x], {x, -2 π, 2 π}, Frame → True, Background → LightPurple]
```



2.3. Matrix of graphs

▼ GaphicsGrid

It is to plot the matrix of graphs that we have defined

★ GraphicsGrid [{graph11, graph12,...},{graph21, graph22,...},...]

If we want to add a frame to the graph: Frame → True

```
a = Plot[Sin[x], {x, -2 Pi, 2 Pi}, PlotStyle → {Blue, Thickness[0.01]}, PlotLabel → "sin"]
```




```
b = Plot[Cos[x], {x, -2 Pi, 2 Pi},
  PlotStyle → {Purple, Thickness[0.01]}, PlotLabel → "cos"]
```



E. Alberdi Celaya
Mª I. Eguia Ribero
Mª J. González Gómez

```
c = Plot[Tan[x], {x, -Pi, Pi},
  Exclusions → {x == -Pi / 2, Pi / 2}, ExclusionsStyle → Dashing[Medium],
  PlotStyle → {Orange, Thickness[0.01]}, PlotLabel → "tangent"]
```




```
d = Plot[Cot[x], {x, -3.2, 3.2}, PlotStyle → {Green, Thickness[0.01]},
  Exclusions → {-Pi, Pi}, ExclusionsStyle → Dashing[Medium], PlotLabel → "cotangent"]
```




```
GraphicsGrid[{{a, b}, {c, d}}, Frame → True]
```